######################################################
##
## EXERCISE 11
##
######################################################

## Select all resources that are an instance of Person that do not have a death date specified for them
#
SELECT *
WHERE {
  ?person a schema:Person .
  FILTER NOT EXISTS {?person schema:deathDate ?date }
}

## List all persons, and if specified, their birthdate and death date
#
SELECT ?person ?birthDate ?deathDate
WHERE {
  ?person a schema:Person .
  OPTIONAL { ?person schema:birthDate ?birthDate } 
  OPTIONAL { ?person schema:deathDate ?deathDate } 
}

## Selects persons who are Kennedys, the university or college they graduated from, their spouse, and, optionally, the university or college their spouse graduated from.
#
SELECT ?person ?univ ?spouse ?spouseUniv
WHERE {
  ?person schema:alumniOf ?univ .
  ?person schema:spouse ?spouse .
  ?person schema:familyName ?familyName .
  FILTER ( ?familyName = "Kennedy" ) 
  OPTIONAL {
    ?spouse schema:alumniOf ?spouseUniv
  }
} 

## Selects persons who are Kennedys, and their spouse, and either the Kennedy graduated from Yale or the spouse graduated from Manhattanville College
#
SELECT ?person ?univ ?spouse ?spouseUniv
WHERE {
  ?person schema:spouse ?spouse .
  ?person schema:familyName ?familyName .
  FILTER ( ?familyName = "Kennedy" ) 
  {
    ?person schema:alumniOf ?univ .
    FILTER (?univ = kennedys:Yale )
  } UNION {
    ?spouse schema:alumniOf ?spouseUniv .
    FILTER ( ?spouseUniv = kennedys:ManhattanvilleCollege )
  }
} 

## Selects all descendants of Rose Fitzgerald that have either navy, attorney or actor as job title.
#
SELECT ?person ?jobTitle
WHERE {
  ?person schema:parent+ kennedys:RoseFitzgerald .
  { 
    ?person schema:jobTitle ?jobTitle .
    FILTER ( ?jobTitle = "navy" ) 
  } UNION {
    ?person schema:jobTitle ?jobTitle .
    FILTER ( ?jobTitle = "attorney" )
  } UNION {
    ?person schema:jobTitle ?jobTitle .
    FILTER ( ?jobTitle = "actor" )
  }
}

## Bonus exercise. Create a query that selects people that have the job title “attorney”, or have an honorific suffix. 
#
SELECT ?person ?givenName ?familyName ?suffix ?jobTitle
WHERE {
  ?person schema:givenName ?givenName .
  ?person schema:familyName ?familyName .
  {
    ?person schema:jobTitle ?jobTitle .
    FILTER ( ?jobTitle = "attorney" )
    FILTER NOT EXISTS {
      ?person schema:honorificSuffix ?suffix .
    }
  } UNION {
    ?person schema:honorificSuffix ?suffix .
    FILTER NOT EXISTS {
      ?person schema:jobTitle "attorney" .
    }
  } UNION {
    ?person schema:honorificSuffix ?suffix .
    ?person schema:jobTitle ?jobTitle .
    FILTER ( ?jobTitle = "attorney" )
  }
} ORDER BY ?familyName ?givenName
